

KRR Languages: Quick Reference Ver. 5.3 (Feb 2024)

S. Baskaran

Introduction

This document outlines the specifications for six languages used in assignments for Knowledge Representation and Reasoning course offered by Prof. Deepak Khemani at IIT Madras. Abstract grammars for the languages are specified in BNF (Backus-Naur Form) notation; keywords, identifiers, constants, and operators (their precedence and associativity) are specified separately.

In the grammars, *non-terminal symbols are in italic font* and **terminal symbols are in blue monospace font** and BNF meta-characters include:

\rightarrow	defines a rule	$?$	zero or one
$ $	alternation	$*$	zero or more
ϵ	empty string	$+$	one or more
$\langle \dots \rangle$	grouping	\triangleright	line comment

Keywords and operators are drawn from ASCII set, additionally, equivalent Unicode math operators are supported to express code in math notation.

The abstract grammars only depict ASCII keywords and operators, the equivalent Unicode math operators are listed separately. Take the Unicode math operators from the sample input files provided the bundle and *do not copy* Unicode operators from this PDF.

1 Propositional Logic

1.1 Grammar

1:	$program \rightarrow \langle sentence \mid block \rangle^*$	
2:	$block \rightarrow \{ \langle sentence \mid block \rangle^* \} \triangleright \{ \alpha; \beta; \{ \gamma; \} \}$	
3:	$sentence \rightarrow formula ;$	$\triangleright \alpha;$
4:	$formula \rightarrow true$	$\triangleright \top$
5:	$ false$	$\triangleright \perp$
6:	$ VARIABLE$	
7:	$ [formula]$	
8:	$ (formula)$	
9:	$ not formula$	$\triangleright \neg \alpha$
10:	$ formula and formula$	$\triangleright \alpha \wedge \beta$
11:	$ formula or formula$	$\triangleright \alpha \vee \beta$
12:	$ formula \Rightarrow formula$	$\triangleright \alpha \supset \beta$
13:	$ formula \Leftarrow formula$	$\triangleright \alpha \subset \beta$
14:	$ formula \Leftrightarrow formula$	$\triangleright \alpha \equiv \beta$

1.2 Symbols

Keywords: **true** (\top), **false** (\perp), **and** (\wedge), **or** (\vee), **not** (\neg).

User defined symbols:

Symbol	Syntax	Example
VARIABLE	<code>[a-zA-Z][A-Za-z0-9_]</code>	H20, k2, sun

1.3 Operators

Listed from highest to lowest precedence and equal precedence listed together.

Arity	Operators	Assoc.
unary	not , \sim , \neg	right
binary	and , \wedge	left
binary	or , \vee	left
binary	=> , \supset	right
binary	<= , \subset	left
binary	<=> , \equiv	right

1.4 Example

```
# This is a line comment.
# Statements end with a semicolon;
```

```
# operator associativity
```

```
[not not P] <=> [not (not P)] ;
[P and Q and R] <=> [(P and Q) and R] ;
[P or Q or R] <=> [(P or Q) or R] ;
[P => Q => R] <=> [P => (Q => R)] ;
[P <= Q <= R] <=> [(P <= Q) <= R] ;
[P <=> Q <=> R] <=> [P <=> (Q <=> R)] ;
```

```
# operator precedence
```

```
[not P and Q] <=> [(not P) and Q] ;
[P or Q and R] <=> [P or (Q and R)] ;
[P => Q or R] <=> [P => (Q or R)] ;
[P => Q <= R] <=> [(P => Q) <= R] ;
[P <=> Q <= R] <=> [P <=> (Q <= R)] ;
```

```
# a block of statements
```

```
{ ~A or B; ~B or ~C or D; C; }
```

2 First Order Logic

2.1 Grammar

- 1: $program \rightarrow \langle sentence \mid block \rangle^*$
- 2: $block \rightarrow \{ \langle sentence \mid block \rangle^* \} \triangleright \{ \alpha; \beta; \{ \gamma; \} \}$
- 3: $sentence \rightarrow formula ; \triangleright \alpha;$
- 4: $formula \rightarrow true \triangleright \top$
- 5: $\quad \mid false \triangleright \perp$
- 6: $\quad \mid term < term \triangleright \text{predicate}$
- 7: $\quad \mid term \leq term \triangleright \text{predicate}$
- 8: $\quad \mid term = term \triangleright \text{predicate}$
- 9: $\quad \mid term \geq term \triangleright \text{predicate}$
- 10: $\quad \mid term > term \triangleright \text{predicate}$
- 11: $\quad \mid term \ltimes term \triangleright \text{predicate}$
- 12: $\quad \mid NAME (terms) \triangleright \text{predicate}$
- 13: $\quad \mid forall vars formula \triangleright \forall x \alpha$
- 14: $\quad \mid exists vars formula \triangleright \exists x \alpha$
- 15: $\quad \mid exactlyone vars formula \triangleright \exists! x \alpha$
- 16: $\quad \mid (formula)$
- 17: $\quad \mid not formula \triangleright \neg \alpha$
- 18: $\quad \mid formula \text{ and } formula \triangleright \alpha \wedge \beta$
- 19: $\quad \mid formula \text{ or } formula \triangleright \alpha \vee \beta$
- 20: $\quad \mid formula \Rightarrow formula \triangleright \alpha \supset \beta$
- 21: $\quad \mid formula \leq formula \triangleright \alpha \subset \beta$
- 22: $\quad \mid formula \leq formula \triangleright \alpha \equiv \beta$
- 23: $vars \rightarrow VARIABLE \langle , VARIABLE \rangle^*$
- 24: $terms \rightarrow term \langle , term \rangle^*$
- 25: $term \rightarrow expr \mid list \triangleright \text{term}$
- 26: $expr \rightarrow INTEGER \triangleright \text{constant}$
- 27: $\quad \mid FLOAT \triangleright \text{constant}$
- 28: $\quad \mid STRING \triangleright \text{constant}$
- 29: $\quad \mid CONSTANT \triangleright \text{constant}$
- 30: $\quad \mid NAME (terms?) \triangleright \text{function}$
- 31: $\quad \mid VARIABLE \triangleright \text{variable}$
- 32: $\quad \mid (expr)$
- 33: $\quad \mid - expr$
- 34: $\quad \mid expr * expr$
- 35: $\quad \mid expr / expr$
- 36: $\quad \mid expr \% expr$
- 37: $\quad \mid expr + expr$
- 38: $\quad \mid expr - expr$
- 39: $list \rightarrow [] \triangleright \text{nil list}$
- 40: $\quad \mid [terms] \triangleright \text{list}$
- 41: $\quad \mid [terms \mid list] \triangleright \text{list}$
- 42: $\quad \mid [terms \mid VARIABLE] \triangleright \text{list}$

2.2 Symbols

Keywords: **true** (\top), **false** (\perp), **forall** (\forall), **exists** (\exists), **exactlyone** ($\exists!$), **and** (\wedge), **or** (\vee), **not** (\neg).

User defined symbols:

Symbol	Syntax	Example
NAME	<code>[a-z][A-Za-z0-9_]*</code>	likes, age
CONSTANT	<code>[a-z][A-Za-z0-9_]*</code>	anna, elsa
STRING	<code>'...'</code>	'WALL-E'
VARIABLE	<code>[A-Z][A-Za-z0-9_]*</code>	X, Y, Z

2.3 Operators

Listed from highest to lowest precedence and equal precedence listed together.

Arity	Operators	Assoc.
unary	<code>-</code>	right
binary	<code>*, /, %</code>	left
binary	<code>+, -</code>	left
binary	<code><, <=, =, >=, >, <></code>	N/A
unary	<code>not, ~, ¬, forall, ∀, exists, ∃, exactlyone, ∃!</code>	right
binary	<code>and, ∧</code>	left
binary	<code>or, ∨</code>	left
binary	<code>=>, ⊃</code>	right
binary	<code><=, ⊂</code>	left
binary	<code><=>, ≡</code>	right

2.4 Example

This is a line comment.

Statements end with a semicolon;

```
(forall X,Y p(X,Y))
  <=> (forall X forall Y p(X,Y))
  <=> (forall X (forall Y p(X,Y))) ;
```

```
[1,2,3] = [1 | [2 | [3 | []] ] ] ;
[1,2,3] = [1,2 | [3 | []] ] ;
[1,2,3] = [1,2,3 | [] ] ;
```

```
(exists X p(X) and q(X))
  <=> ((exists X p(X)) and q(X)) ;
```

```
2+3-4*5/10 = (2+3)-((4*5)/10) ;
```

```
{
  forall X ( man(X) => mortal(X) ) ;
  forall X ( mortal(X) <= man(X) ) ;
  true and not exists X ( p(X) and not q(x) ) ;
  forall X,Y,Z ( p(X,Y) and p(Y,Z) => p(X,Z) ) ;
}
```

3 Horn Clauses

Horn clause syntax is borrowed from SWI Prolog, with tiny differences. And it is easy to convert horn clause programs into SWI Prolog programs.

3.1 Grammar

- 1: *program* → *hornClause**
- 2: *hornClause* → *predicate* :- *body* . ▷ rule
- 3: | *predicate* . ▷ fact
- 4: | *body* ? ▷ query
- 5: *predicate* → **NAME** (*terms*)
- 6: *body* → *subgoal* ⟨ , *subgoal* ⟩*
- 7: *subgoal* → ! ▷ cut operator
- 8: | *literal*
- 9: | ⟨ ~ | **not** | \+ ⟩ *literal*
- 10: *literal* → **true** | **false**
- 11: | *term* < *term* ▷ eval., compare
- 12: | *term* <= *term* ▷ eval., compare
- 13: | *term* >= *term* ▷ eval., compare
- 14: | *term* > *term* ▷ eval., compare
- 15: | *term* **is** *term* ▷ eval., compare
- 16: | *term* **==** *term* ▷ eval., compare
- 17: | *term* **=\=** *term* ▷ eval., compare
- 18: | *term* **=** *term* ▷ unifies with
- 19: | *term* **\=** *term* ▷ not unifies with
- 20: | *term* **==** *term* ▷ equivalent to
- 21: | *term* **\==** *term* ▷ not equivalent to
- 22: | *predicate*
- 23: | (*literal*)
- 24: *terms* → *term* ⟨ , *term* ⟩*
- 25: *term* → *expr* | *list*
- 26: *expr* → **INTEGER** ▷ constant
- 27: | **FLOAT** ▷ constant
- 28: | **STRING** ▷ constant
- 29: | **CONSTANT** ▷ constant
- 30: | **NAME** (*terms*?) ▷ function
- 31: | **VARIABLE** ▷ variable
- 32: | (*expr*)
- 33: | - *expr*
- 34: | *expr* * *expr*
- 35: | *expr* / *expr*
- 36: | *expr* % *expr*
- 37: | *expr* + *expr*
- 38: | *expr* - *expr*
- 39: *list* → [] ▷ nil list
- 40: | [*terms*] ▷ list
- 41: | [*terms* | *list*] ▷ list
- 42: | [*terms* | **VARIABLE**] ▷ list

3.2 Symbols

Keywords: **true**, **false**, **not**, **is**.

User defined symbols:

Symbol	Syntax	Example
NAME	[a-z][A-Za-z0-9_]*	likes, age
CONSTANT	[a-z][A-Za-z0-9_]*	anna, elsa
STRING	'...'	'WALL-E'
VARIABLE	[A-Z][A-Za-z0-9_]*	A, B, X, Y

3.3 Operators

Listed from highest to lowest precedence and equal precedence listed together.

Arity	Operators	Assoc.
unary	-	right
binary	*, /, %	left
binary	+, -	left
binary	<, <=, >=, >, is , == , =\= , = , \= , == , \==	N/A
unary	~, not , \+	right

3.4 Example

```
# This is a line comment.
# Statements end with a dot.

# append([1], [2], [1,2]).

append([], B, B).
append([X|A], B, [X|C]) :- append(A, B, C).

append([1,2],[3,4],[1,2,3,4]) ? # query
!, \+ append([1,2],[3,4],C) ? # query

parent(P,X)      :- mother(P,X).
parent(P,X)      :- father(P,X).
grandparent(G,X) :- parent(G,P), parent(P,X).

cousin(X,Y)      :- X \= Y,
                  \+ sibling(X,Y),
                  grandparent(Z,X),
                  grandparent(Z,Y).

americanCousin(X,Y) :- cousin(X,Y), !,
                  american(X).

composite(N) :- N > 1, ~ prime(N).
composite(N) :- N > 1, \+ prime(N).
composite(N) :- N > 1, not (prime(N)).
```

4 Production Systems

The working-memory-elements (WMEs) and rules follow the syntax given in “Knowledge Representation and Reasoning” by Brachman and Levesque. We have made some enhancements to make it expressive. And it allows **insert** as an alias for **add** action.

4.1 Grammar

- 1: $program \rightarrow \langle wme \mid rule \rangle^*$
- 2: $wme \rightarrow (\text{TYPE } attrSpec^*) ;$
- 3: $attrSpec \rightarrow \text{ATTRIBUTE} : evalExpr$
- 4: $rule \rightarrow \text{if } condition + \text{then } action + ;$
- 5: $condition \rightarrow wmeTest \mid - wmeTest$
- 6: $action \rightarrow \langle \text{add} \mid \text{insert} \rangle wme$
- 7: $\quad \mid \text{remove INTEGER}$
- 8: $\quad \mid \text{modify INTEGER } (attrSpec^*)$
- 9: $wmeTest \rightarrow (\text{TYPE } \langle \text{ATTRIBUTE} : testSpec \rangle^*)$
- 10: $testSpec \rightarrow \text{null} \quad \triangleright \text{undefined attribute}$
- 11: $\quad \mid - \quad \triangleright \text{defined attribute}$
- 12: $\quad \mid evalExpr \text{ testExpr?}$
- 13: $\quad \mid testExpr$
- 14: $testExpr \rightarrow \{ test \} \quad \triangleright \text{enclose in curly braces}$
- 15: $test \rightarrow cOP \text{ evalExpr} \quad \triangleright \text{relational}$
- 16: $\quad \mid (test) \quad \triangleright \text{boolean}$
- 17: $\quad \mid \text{not } test \quad \triangleright \text{boolean}$
- 18: $\quad \mid test \text{ bOP } test \quad \triangleright \text{boolean}$
- 19: $evalExpr \rightarrow atom$
- 20: $\quad \mid [expr] \quad \triangleright \text{enclose in sq. brackets}$
- 21: $expr \rightarrow atom$
- 22: $\quad \mid (expr)$
- 23: $\quad \mid - expr \quad \triangleright \text{arithmetic}$
- 24: $\quad \mid \text{not } expr \quad \triangleright \text{boolean}$
- 25: $\quad \mid expr \text{ aOP } expr \quad \triangleright \text{arithmetic}$
- 26: $\quad \mid expr \text{ cOP } expr \quad \triangleright \text{relational}$
- 27: $\quad \mid expr \text{ bOP } expr \quad \triangleright \text{boolean}$
- 28: $atom \rightarrow \text{true} \quad \triangleright \text{constant}$
- 29: $\quad \mid \text{false} \quad \triangleright \text{constant}$
- 30: $\quad \mid \text{INTEGER} \quad \triangleright \text{constant}$
- 31: $\quad \mid \text{FLOAT} \quad \triangleright \text{constant}$
- 32: $\quad \mid \text{STRING} \quad \triangleright \text{constant: 'John'}$
- 33: $\quad \mid \text{CONSTANT} \quad \triangleright \text{constant: john}$
- 34: $\quad \mid \text{VARIABLE} \quad \triangleright \text{variable: X, Y}$
- 35: $aOP \rightarrow * \mid / \mid \% \mid + \mid - \quad \triangleright \text{arithmetic}$
- 36: $cOP \rightarrow < \mid <= \mid = \mid >= \mid > \mid <> \quad \triangleright \text{relational}$
- 37: $bOP \rightarrow \text{and} \mid \text{or} \quad \triangleright \text{boolean}$

4.2 Symbols

Keywords: **if**, **then**, **add**, **insert**, **modify**, **remove**, **true**, **false**, **and**, **or**, **not**, **null**, **'_'**; where a **null** denotes an undefined attribute, and an underscore **'_'** denotes a defined attribute.

User defined symbols:

Symbol	Syntax	Example
TYPE	[a-z][A-Za-z0-9_-]*	car, bus
ATTRIBUTE	[a-z][A-Za-z0-9_-]*	hue, size
CONSTANT	[a-z][A-Za-z0-9_-]*	red, big
STRING	'...'	'WALL-E'
VARIABLE	[A-Z][A-Za-z0-9_-]*	A, X, Y

4.3 Operators

Listed from highest to lowest precedence and equal precedence listed together.

Arity	Operators	Assoc.
unary	-	right
binary	*, /, %	left
binary	+, -	left
binary	<, <=, =, >=, >, <>	N/A
unary	~, not	right
binary	and	left
binary	or	left

4.4 Example

This is a line comment.

Statements end with a semicolon;

```
(phone item:a15 color:red mem:[32+32] camera:2) ;
(stock item:a15 qty:10) ;
```

```
IF (phone item: X
    color: {=red or =blue}
    memory: {>= 32 and <= 128}
    cost: MRP {<= 10 or >= 20}
    tax: TAX {<= [MRP*5/100]}
    warranty: _ # defined attr.
    touchScreen: null # undefined attr.
)
(stock item:X qty:Q {> 0} )
(delivery item:X days: {not > 5} )
(loan item:X emi:E {< [(MRP-TAX)/10]})
-(insurance item:X cost: {>= [E/2]} )
THEN
REMOVE 1
MODIFY 2 (qty: [Q - 1])
ADD (cart item:X price:MRP qty:1)
;
```

5 Description Logic: \mathcal{DL}

This is an implementation of the description logic language \mathcal{DL} from “Knowledge Representation and Reasoning” by Brachman and Levesque.

5.1 Grammar

- 1: $program \rightarrow \langle sentence \mid block \rangle^*$
- 2: $block \rightarrow \{ \langle sentence \mid block \rangle^* \} \triangleright \{ \alpha; \beta; \{ \gamma; \} \}$
- 3: $sentence \rightarrow concept \Rightarrow concept ; \quad \triangleright A \sqsubseteq B;$
- 4: $\quad \mid concept = concept ; \quad \triangleright A \doteq B;$
- 5: $\quad \mid symbols \rightarrow concept ; \quad \triangleright a, b \rightarrow A;$
- 6: $symbols \rightarrow symbol \langle , symbol \rangle^*$
- 7: $symbol \rightarrow \text{STRING} \mid \text{CONSTANT}$
- 8: $concept \rightarrow \text{NAME} \quad \triangleright \text{atomic concept}$
- 9: $\quad \mid [\text{fills role constant}]$
- 10: $\quad \mid [\text{all role concept}]$
- 11: $\quad \mid [\text{exists INTEGER role}]$
- 12: $\quad \mid [\text{and concept concept} +]$
- 13: $role \rightarrow : \text{NAME}$
- 14: $constant \rightarrow \text{INTEGER}$
- 15: $\quad \mid \text{FLOAT}$
- 16: $\quad \mid \text{STRING}$
- 17: $\quad \mid \text{CONSTANT}$
- 18: $\quad \mid (constant)$
- 19: $\quad \mid - constant$
- 20: $\quad \mid constant aOP constant$
- 21: $aOP \rightarrow * \mid / \mid \% \mid + \mid -$

5.2 Symbols

Keywords: **fills**, **all**, **exists**, **and**

User defined symbols:

Symbol	Syntax	Example
NAME	<code>[A-Z][A-Za-z0-9_-]*</code>	Man, Mortal
CONSTANT	<code>[a-z][A-Za-z0-9_-]*</code>	anna, elsa
STRING	<code>'...'</code>	'WALL-E'

5.3 Operators

Listed from highest to lowest precedence and equal precedence listed together.

Arity	Operators	Assoc.
unary	<code>-</code>	right
binary	<code>*, /, %</code>	left
binary	<code>+, -</code>	left
binary	<code>=>, =, ->, $\sqsubseteq, \doteq, \rightarrow$</code>	N/A

5.4 Example

This is a line comment.
Statements end with a semicolon;

Surgeon => Doctor ;

```
BlendedRedWine =
    [AND Wine
      [FILLS :Color red]
      [EXISTS 2 :GrapeType]
    ] ;
```

a block of statements

```
{
    ProgressiveCompany =
        [AND Company
          [EXISTS 7 :Director]
          [ALL :Manager
            [AND Woman
              [FILLS :Degree PhD] ] ]
          [FILLS :MinSalary 24.00/hour]
        ] ;

    joe -> Person ;

    canCorp ->
        [AND Company
          [ALL :Manager Canadian]
          [FILLS :Manager joe]
        ] ;
}
```

6 Subset of OWL 2

This is a subset of OWL 2, it follows OWL Manchester Syntax¹ from “OWL 2 Web Ontology Language Manchester Syntax (Second Edition) 11 Dec. 2012”².

In the grammar³, the meta rule $\alpha LIST$ denotes a comma separated list of α values, similarly, $\alpha 2LIST$ denotes a list of two or more α values.

6.1 Grammar

```

1: kb → frame*

2: frame → classFrame
3:       | roleFrame
4:       | individualFrame
5:       | EquivalentClasses: description2LIST
6:       | DisjointClasses: description2LIST
7:       | EquivalentProperties: role2LIST
8:       | DisjointProperties: role2LIST
9:       | SameIndividual: individual2LIST
10:      | DifferentIndividuals: individual2LIST

11: classFrame → Class: NAME
12:   < SubClassOf: descriptionLIST
13:     | EquivalentTo: descriptionLIST
14:     | DisjointWith: descriptionLIST
15:     | DisjointUnionOf: description2LIST
16:   >*

17: roleFrame → ObjectProperty: NAME
18:   < Domain: descriptionLIST
19:     | Range: descriptionLIST
20:     | Characteristics: characteristicLIST
21:     | SubPropertyOf: roleExprLIST
22:     | EquivalentTo: roleExprLIST
23:     | DisjointWith: roleExprLIST
24:     | InverseOf: roleExprLIST
25:     | SubPropertyChain: roleChain
26:   >*

27: roleExpr → ROLE | inverse ROLE

28: roleChain → roleExpr < o roleExpr >+

29: characteristic → Functional
30:                 | InverseFunctional
31:                 | Reflexive
32:                 | Irreflexive
33:                 | Symmetric
34:                 | Asymmetric
35:                 | Transitive

```

```

36: individualFrame → Individual: NAME
37:   < Types: descriptionLIST
38:     | Facts: factLIST
39:     | SameAs: individualLIST
40:     | DifferentFrom: individualLIST
41:   >*

```

```

42: fact → ROLE INDIVIDUAL
43:     | not ROLE INDIVIDUAL

```

```

44: description → CONCEPT
45:   | roleExpr only description
46:   | roleExpr some description
47:   | roleExpr value description
48:   | roleExpr min INTEGER description
49:   | roleExpr max INTEGER description
50:   | roleExpr exactly INTEGER description
51:   | not description
52:   | ( description )
53:   | description and description
54:   | description or description

```

6.2 Symbols

Keywords (in OWL Manchester Syntax) are **case sensitive**: **not**, **and**, **or**, **inverse**, **only**, **some**, **min**, **max**, **exactly**, **value**, **o**, **Functional**, **InverseFunctional**, **Reflexive**, **Irreflexive**, **Symmetric**, **Asymmetric**, **Transitive**, **Prefix**, **Ontology**, **Class**, **SubClassOf**, **EquivalentTo**, **DisjointWith**, **DisjointUnionOf**, **ObjectProperty**, **Characteristics**, **Domain**, **Range**, **SubPropertyOf**, **InverseOf**, **SubPropertyChain**, **Individual**, **Types**, **Facts**, **SameAs**, **DifferentFrom**, **EquivalentClasses**, **DisjointClasses**, **EquivalentProperties**, **DisjointProperties**, **SameIndividual**, **DifferentIndividuals**.

User defined symbols:

Symbol	Syntax	Example
NAME	[a-zA-Z][A-Za-z0-9_-]*	owns, Car
CONCEPT	NAME	Car, Bus
ROLE	NAME	owns, eats
INDIVIDUAL	NAME	Lucy, Jack

6.3 Operators

Operators from highest to lowest precedence.

Arity	Operators	Assoc.
unary	not	N/A
binary	and	left
binary	or	left

¹With suitable **Prefix**: and **Ontology**: entries, these files can be opened in Protégé ontology management tool.

²<https://www.w3.org/TR/owl2-manchester-syntax/>

³A frame is a block of statements, it differs from Frames discussed in Brachman and Levesque, but has similar syntax.

6.4 Example

This is a line comment.

Class: Person

SubClassOf: eats some Fruit

EquivalentTo: Human

DisjointWith: Fruit, Meat

DisjointUnionOf: Man, Woman

Class: TOAD

EquivalentTo: Teen and owns some Apple

SubClassOf: Happy

ObjectProperty: hasChild

Domain: Person

Range: Person

InverseOf: hasParent

ObjectProperty: hasSibling

Characteristics: Symmetric

Domain: Person

Range: Person

ObjectProperty: hasBrother

Domain: Person

Range: Man

SubPropertyOf: hasSibling

ObjectProperty: hasSister

Domain: Person

Range: Woman

SubPropertyOf: hasSibling

Individual: Lucy

Types: Woman, hasChild only Woman

Facts: hasHusband Manny, not owns Car157

SameAs: SmartLucy

DifferentFrom: Manny, Car157

EquivalentClasses: Dead, not Alive

DisjointClasses: Fruit, Meat

EquivalentProperties: owns, hasOwnershipOf

DisjointProperties: hasBrother, hasSister

SameIndividual: Manny, LazyManny

DifferentIndividuals: Manny, Diego, Sid, Lucy

6.5 Note

The subset of OWL 2 that we implemented is a superset of \mathcal{ALC} , for example, the subset supports min/max cardinality restrictions, inverse roles, etc., that are not supported in \mathcal{ALC} . While testing \mathcal{ALC} Tableau use only \mathcal{ALC} constructs discussed in the lecture.

6.6 \mathcal{ALC} Examples

Persons who do not own a car.

$\text{Person} \sqcap \neg \exists \text{owns. Car}$

Person and not (owns some Car)

Those who do not travel by bus or train.

$\neg \exists \text{travelsBy. (Bus} \sqcup \text{Car)}$

not (travelsBy some (Bus or Train))

$\forall \text{travelsBy.} \neg (\text{Bus} \sqcup \text{Car})$

travelsBy only not (Bus or Train)

Owns a thing that has battery.

$\exists \text{owns.} (\exists \text{hasPart. Battery})$

owns some (hasPart some Battery)

Those with friends who own only electric cars.

$\exists \text{hasFriend.} (\forall \text{owns. (Electric} \sqcap \text{Car)})$

hasFriend some (owns only (Electric and Car))

Lucy is a mother and an engineer, she works for Acme Co. Her brother Jack is a doctor and he owns a car.

Mother(Lucy), Engineer(Lucy),

worksFor(Lucy, AcmeCo), hasBrother(Lucy, Jack),

Doctor(Jack), $(\exists \text{owns. Car})(\text{Jack})$

Individual: Lucy

Types: Mother, Engineer

Facts: worksFor AcmeCo, hasBrother Jack

Individual: Jack

Types: Doctor, owns some Car

Domain of hasBrother is Person and range is Man.

$\exists \text{hasBrother.} \top \sqsubseteq \text{Person}$ (domain axiom)

$\top \sqsubseteq \forall \text{hasBrother. Man}$ (range axiom)

ObjectProperty: hasBrother

Domain: Person

Range: Man